NAG Fortran Library Routine Document

G13FAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G13FAF estimates the parameters of either a standard univariate regression GARCH process, or a univariate regression-type I AGARCH(p, q) process (see Engle and Ng (1993)).

2 Specification

```
 SUBROUTINE G13FAF(DIST, YT, X, LDX, NUM, IP, IQ, NREG, MN, ISYM, NPAR,
1                  THETA, SE, SC, COVAR, LDC, HP, ET, HT, LGF, COPTS,
2                  MAXIT, TOL, WORK, LWORK, IFAIL)
 INTEGER           LDX, NUM, IP, IQ, NREG, MN, ISYM, NPAR, LDC, MAXIT,
1                  LWORK, IFAIL
 real              YT(NUM), X(LDX,*), THETA(NPAR), SE(NPAR), SC(NPAR),
1                  COVAR(LDC,NPAR), HP, ET(NUM), HT(NUM), LGF, TOL,
2                  WORK(LWORK)
 LOGICAL           COPTS(2)
 CHARACTER*1       DIST
```

3 Description

A univariate regression-type I AGARCH(p, q) process, with q coefficients α_i, for $i = 1, \ldots, q$, p coefficients β_i, for $i = 1, \ldots, p$, and k linear regression coefficients b_i, for $i = 1, \ldots, k$, can be represented by:

$$y_t = b_o + {x_t}^T b + \epsilon_t \tag{1}$$

$$h_t = \alpha_0 + \sum_{i=1}^{q} \alpha_i (\epsilon_{t-i} + \gamma)^2 + \sum_{i=1}^{p} \beta_i h_{t-i}, \quad t = 1, \ldots, T \tag{2}$$

where $\epsilon_t | \psi_{t-1} = N(0, h_t)$ or $\epsilon_t | \psi_{t-1} = S_t(df, h_t)$. Here S_t is a standardised Student's t-distribution with df degrees of freedom and variance h_t, T is the number of terms in the sequence, y_t denotes the endogenous variables, x_t the exogenous variables, b_o the regression mean, b the regression coefficients, ϵ_t the residuals, h_t is the conditional variance, df the number of degrees of freedom of the Student's t-distribution, and ψ_t the set of all information up to time t.

G13FAF provides an estimate for $\hat{\theta}$, the parameter vector $\theta = (b_o, b^T, \omega^T)$ where $b^T = (b_1, \ldots, b_k)$, $\omega^T = (\alpha_0, \alpha_1, \ldots, \alpha_q, \beta_1, \ldots, \beta_p, \gamma)$ when DIST = 'N' and $\omega^T = (\alpha_0, \alpha_1, \ldots, \alpha_q, \beta_1, \ldots, \beta_p, \gamma, df)$ when DIST = 'T'.

ISYM, MN and NREG can be used to simplify the GARCH(p, q) expression in (1) as follows:

No Regression and No Mean

$y_t = \epsilon_t$,

ISYM $= 0$,

MN $= 0$,

NREG $= 0$ and

θ is a $(p + q + 1)$ vector, when DIST = 'N' and a $(p + q + 2)$ vector, when DIST = 'T'.

No Regression

$y_t = b_o + \epsilon_t$,

ISYM $= 0$,

MN $= 1$,

NREG $= 0$ and

θ is a $(p + q + 2)$ vector when DIST = 'N' and a $(p + q + 3)$ vector, when DIST = 'T'.

Note: if the $y_t = \mu + \epsilon_t$, where μ is known (not to be estimated by G13FAF) then (1) can be written as ${y_t}^{\mu} = \epsilon_t$, where ${y_t}^{\mu} = y_t - \mu$. This corresponds to the case **No Regression and No Mean**, with y_t replaced by $y_t - \mu$.

No Mean

$y_t = {x_t}^T b + \epsilon_t$,

ISYM $= 0$,

MN $= 0$,

NREG $= k$ and

θ is a $(p + q + k + 1)$ vector when DIST = 'N' and a $(p + q + k + 2)$ vector, when DIST = 'T'.

4 References

Engle R (1982) Autoregressive conditional heteroskedasticity with estimates of the variance of United Kingdom inflation *Econometrica* **50** 987–1008

Bollerslev T (1986) Generalised autoregressive conditional heteroskedasticity *Journal of Econometrics* **31** 307–327

Engle R and Ng V (1993) Measuring and Testing the Impact of News on Volatility *Journal of Finance* **48** 1749–1777

Hamilton J (1994) *Time Series Analysis* Princeton University Press

5 Parameters

1: DIST – CHARACTER*1 *Input*

On entry: the type of distribution to use for e_t.

If DIST = 'N', a Normal distribution is used.

If DIST = 'T', a Student's t-distribution is used.

Constraint: DIST = 'N' or 'T'.

2: YT(NUM) – ***real*** array *Input*

On entry: the sequence of observations, y_t, for $t = 1, \ldots, T$.

3: X(LDX,*) – ***real*** array *Input*

Note: the second dimension of the array X must be at least $\max(1, \mathrm{NREG} + \mathrm{MN})$.

On entry: row t of X must contain the time dependent exogenous vector x_t, where ${x_t}^T = ({x_t}^1, \ldots, {x_t}^k)$, for $t = 1, \ldots, T$.

4: LDX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which G13FAF is called.

Constraint: LDX $\geq$ NUM.

5: NUM – INTEGER *Input*

On entry: the number of terms in the sequence, T.

Constraint: NUM $\geq \max(\text{IP}, \text{IQ})$.

6: IP – INTEGER *Input*

On entry: the number of coefficients, β_i, for $i = 1, \ldots, p$.

Constraint: IP ≥ 0 (see also NPAR).

7: IQ – INTEGER *Input*

On entry: the number of coefficients, α_i, for $i = 1, \ldots, q$.

Constraint: IQ ≥ 1 (see also NPAR).

8: NREG – INTEGER *Input*

On entry: the number of regression coefficients, k.

Constraint: NREG ≥ 0 (see also NPAR).

9: MN – INTEGER *Input*

On entry: if MN $= 1$, the mean term b_0 will be included in the model.

Constraint: MN $= 0$ or 1.

10: ISYM – INTEGER *Input*

On entry: if ISYM $= 1$ then the asymmetry term γ will be included in the model.

Constraint: ISYM $= 0$ or 1.

11: NPAR – INTEGER *Input*

On entry: the number of parameters to be included in the model. NPAR $= 1 +$ IQ $+$ IP $+$ ISYM $+$ MN $+$ NREG when DIST $=$ 'N', and NPAR $= 2 +$ IQ $+$ IP $+$ ISYM $+$ MN $+$ NREG when DIST $=$ 'T'.

Constraint: NPAR < 20.

12: THETA(NPAR) – ***real*** array *Input/Output*

On entry: the initial parameter estimates for the vector θ. The first element must contain the coefficient α_o and the next IQ elements must contain the coefficients α_i, for $i = 1, \ldots, q$. The next IP elements must contain the coefficients β_j, for $j = 1, \ldots, p$. If ISYM $= 1$, then the next element must contain the asymmetry parameter γ. If DIST $=$ 'T', then the next element must contain df, the number of degrees of freedom of the Student's t-distribution. If MN $= 1$, then the next element must contain the mean term b_o. If COPTS(2) $=$ .FALSE., then the remaining NREG elements are taken as initial estimates of the linear regression coefficients b_i, for $i = 1, \ldots, k$.

On exit: the estimated values $\hat{\theta}$ for the vector θ. The first element contains the coefficient α_o, the next IQ elements contain the coefficients α_i, for $i = 1, \ldots, q$. The next IP elements are the coefficients β_j, for $j = 1, \ldots, p$. If ISYM $= 1$ then the next element contains the estimate for the asymmetry parameter γ. If DIST $=$ 'T' then the next element contains an estimate for df, the number of degrees of freedom of the Student's t-distribution. If MN $= 1$ then the next element contains an estimate for the mean term b_o. The final NREG elements are the estimated linear regression coefficients b_i, for $i = 1, \ldots, k$.

13: SE(NPAR) – ***real*** array *Output*

On exit: the standard errors for $\hat{\theta}$. The first element contains the standard error for α_o. The next IQ elements contain the standard errors for α_i, for $i = 1, \ldots, q$. The next IP elements are the standard errors for β_j, for $j = 1, \ldots, p$. If ISYM $= 1$, then the next element contains the standard error for γ.

If DIST = 'T', then the next element contains the standard error for df, the number of degrees of freedom of the Student's t-distribution. If MN = 1, then the next element contains the standard error for b_o. The final NREG elements are the standard errors for b_j, for $j = 1, \ldots, k$.

14: SC(NPAR) – ***real*** array *Output*

On exit: the scores for $\hat{\theta}$. The first element contains the score for α_o. The next IQ elements contain the score for α_i, for $i = 1, \ldots, q$. The next IP elements are the scores for β_j, for $j = 1, \ldots, p$. If ISYM = 1, then the next element contains the score for γ. If DIST = 'T' then the next element contains the score for df, the number of degrees of freedom of the Student's t-distribution. If MN = 1, then the next element contains the score for b_o. The final NREG elements are the scores for b_j, for $j = 1, \ldots, k$.

15: COVAR(LDC,NPAR) – ***real*** array *Output*

On exit: the covariance matrix of the parameter estimates $\hat{\theta}$, that is the inverse of the Fisher Information Matrix.

16: LDC – INTEGER *Input*

On entry: the first dimension of the array COVAR as declared in the (sub)program from which G13FAF is called.

Constraint: LDC $\geq$ NPAR.

17: HP – ***real*** *Input/Output*

On entry: if COPTS(2) = .FALSE., then HP is the value to be used for the pre-observed conditional variance; otherwise HP is not referenced.

On exit: if COPTS(2) = .TRUE., then HP is the estimated value of the pre-observed of the conditional variance.

18: ET(NUM) – ***real*** array *Output*

On exit: the estimated residuals, ϵ_t, for $t = 1, \ldots, T$.

19: HT(NUM) – ***real*** array *Output*

On exit: the estimated conditional variances, h_t, for $t = 1, \ldots, T$.

20: LGF – ***real*** *Output*

On exit: the value of the log likelihood function at $\hat{\theta}$.

21: COPTS(2) – LOGICAL array *Input*

On entry: the options to be used by G13FAF.

If COPTS(1) = .TRUE., stationary conditions are enforced, otherwise they are not.

If COPTS(2) = .TRUE., the routine provides initial parameter estimates of the regression terms, otherwise these are provided by the user.

22: MAXIT – INTEGER *Input*

On entry: the maximum number of iterations to be used by the optimization routine when estimating the GARCH(p, q) parameters. If MAXIT is set to 0 then the standard errors, score vector and variance-covariance are calculated for the input value of θ in THETA; however the value of θ is not updated.

Constraint: MAXIT $\geq$ 0.

23: TOL – ***real*** *Input*

On entry: the tolerance to be used by the optimization routine when estimating the GARCH(p, q) parameters.

24: WORK(LWORK) – ***real*** array *Workspace*

25: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which G13FAF is called.

Constraint: LWORK $\geq$ (NREG + 3) $\times$ NUM + NPAR + 403.

26: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq$ 0 on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, NREG < 0,
or MN > 1,
or MN < 0,
or ISYM > 1,
or ISYM < 0,
or IQ < 1,
or IP < 0,
or NPAR $\geq$ 20,
or NPAR has an invalid value,
or LDC < NPAR,
or LDX < NUM,
or DIST $\neq$ 'N',
or DIST $\neq$ 'T',
or MAXIT < 0,
or NUM < max(IP, IQ).

IFAIL = 2

On entry, LWORK < (NREG + 3) $\times$ NUM + NPAR + 403.

IFAIL = 3

The matrix X is not full rank.

IFAIL = 4

The information matrix is not positive definite.

IFAIL = 5

The maximum number of iterations has been reached.

IFAIL = 6

The log-likelihood cannot be optimised any further.

IFAIL = 7

No feasible model parameters could be found.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example program uses G05HKF to generate 1500 data points, with known process parameters θ for the following two time-series:

(i) An ARCH(3) sequence with normally distributed residuals.

(ii) A GARCH(1,2) sequence with Student's t-distributed residuals.

Here G05HKF is initially called, with the output discarded, to eliminate 'start-up effects' in these sequences. The process parameter estimates, $\hat{\theta}$, are then obtained using G13FAF, and compared with their true values, θ. Finally a four step ahead volatility estimate is computed using G13FBF.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G13FAF Example Program Text
*     Mark 20 Release. NAG Copyright 2001.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          NPARMX, NUM
      real             ZERO
      PARAMETER        (NPARMX=10,NUM=1500,ZERO=0.0e0)
      INTEGER          NUM1, NREGMX
      PARAMETER        (NUM1=3000,NREGMX=10)
*     .. Local Scalars ..
      real             DF, FAC1, GAMMA, HP, LGF, MEAN, TOL, XTERM
      INTEGER          I, IFLAG, IGEN, IP, IQ, ISYM, K, LDX, LWK, MAXIT,
     +                 MN, NPAR, NPAR2, NREG, NT
      LOGICAL          FCALL
      CHARACTER        DIST
*     .. Local Arrays ..
      real             BX(10), COVAR(NPARMX,NPARMX), CVAR(100),
     +                 ETM(NUM1), HT(NUM1+10), HTM(NUM1), PARAM(NPARMX),
     +                 RVEC(40), RWSAV(9), SC(NPARMX), SE(NPARMX),
     +                 THETA(NPARMX), WK(NUM1*3+NPARMX+NREGMX*NUM1+20*
     +                 20+1), X(NUM1,10), YT(NUM1+10)
      INTEGER          ISEED(4)
      LOGICAL          COPTS(2)
*     .. External Subroutines ..
      EXTERNAL         G05CBF, G05HKF, G05KBF, G13FAF, G13FBF
*     .. Intrinsic Functions ..
      INTRINSIC        real, SIN
```

```
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G13FAF Example Program Results'

      ISEED(1) = 111
      IGEN = 0

      NREG = 0
      LDX = NUM1
      BX(1) = 1.5e0
      BX(2) = 2.5e0
      BX(3) = 3.0e0
      MEAN = 3.0e0
      DO 20 I = 1, NUM
         FAC1 = real(I)*0.01e0
         X(I,1) = 0.01e0 + 0.7e0*SIN(FAC1)
         X(I,2) = 0.5e0 + FAC1*0.1e0
         X(I,3) = 1.0e0
   20 CONTINUE

      ISYM = 1
      MN = 1
      GAMMA = -0.4e0
      IP = 0
      IQ = 3
      PARAM(1) = 0.8e0
      PARAM(2) = 0.6e0
      PARAM(3) = 0.2e0
      PARAM(4) = 0.1e0

      NPAR = 1 + IQ + IP

      LWK = NREG*NUM + 3*NUM + NPAR + ISYM + MN + NREG + 403

      FCALL = .TRUE.
      IFLAG = 0
      DIST = 'N'

      CALL G05KBF(IGEN,ISEED)

      CALL G05HKF(DIST,NUM,IP,IQ,PARAM,GAMMA,DF,HT,YT,FCALL,RVEC,IGEN,
     +            ISEED,RWSAV,IFLAG)

      FCALL = .FALSE.
      CALL G05HKF(DIST,NUM,IP,IQ,PARAM,GAMMA,DF,HT,YT,FCALL,RVEC,IGEN,
     +            ISEED,RWSAV,IFLAG)

      IFLAG = -1

      DO 60 I = 1, NUM
         XTERM = ZERO
         DO 40 K = 1, NREG
            XTERM = XTERM + X(I,K)*BX(K)
   40    CONTINUE
         IF (MN.EQ.1) THEN
            YT(I) = MEAN + XTERM + YT(I)
         ELSE
            YT(I) = XTERM + YT(I)
         END IF
   60 CONTINUE

      COPTS(1) = .TRUE.
      COPTS(2) = .TRUE.

      MAXIT = 200
      TOL = 1.0e-5

      DO 80 I = 1, NPAR
         THETA(I) = PARAM(I)*0.5e0
   80 CONTINUE

      IF (ISYM.EQ.1) THEN
```

```
         THETA(NPAR+ISYM) = GAMMA*0.5e0
      END IF

      IFLAG = 0
      NPAR2 = 1 + IQ + IP + ISYM + MN + NREG
      CALL G13FAF(DIST,YT,X,LDX,NUM,IP,IQ,NREG,MN,ISYM,NPAR2,THETA,SE,
     +            SC,COVAR,NPARMX,HP,ETM,HTM,LGF,COPTS,MAXIT,TOL,WK,LWK,
     +            IFLAG)

      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Normal distribution'
      WRITE (NOUT,*)
      WRITE (NOUT,*) '          Parameter       Standard        Correct'
      WRITE (NOUT,*) '          estimates       errors          values'

      DO 100 I = 1, NPAR
         WRITE (NOUT,99999) THETA(I), SE(I), PARAM(I)
  100 CONTINUE

      IF (ISYM.EQ.1) THEN
         WRITE (NOUT,99999) THETA(NPAR+1), SE(NPAR+1), GAMMA
      END IF

      IF (MN.EQ.1) THEN
         WRITE (NOUT,99999) THETA(NPAR+ISYM+1), SE(NPAR+ISYM+1), MEAN
      END IF

      DO 120 I = 1, NREG
         WRITE (NOUT,99999) THETA(NPAR+ISYM+MN+I), SE(NPAR+ISYM+MN+I),
     +     BX(I)
  120 CONTINUE

      NT = 4
      CALL G13FBF(NUM,NT,IP,IQ,THETA,GAMMA,CVAR,HTM,ETM,IFLAG)

      WRITE (NOUT,*)
      WRITE (NOUT,99998) 'Volatility forecast = ', CVAR(NT)
      WRITE (NOUT,*)


      DIST = 'T'
      NREG = 2
      MN = 1
      DF = 4.1e0
      IP = 1
      IQ = 2
      ISYM = 1
      GAMMA = -0.2e0

      NPAR = IQ + IP + 1
      LWK = NREG*NUM + 3*NUM + NPAR + ISYM + MN + NREG + 404

      PARAM(1) = 0.1e0
      PARAM(2) = 0.2e0
      PARAM(3) = 0.3e0
      PARAM(4) = 0.4e0
      PARAM(5) = 0.1e0

      FCALL = .TRUE.
      CALL G05CBF(111)
      IGEN = 0
      ISEED(1) = 111
      CALL G05KBF(IGEN,ISEED)

      CALL G05HKF(DIST,NUM,IP,IQ,PARAM,GAMMA,DF,HT,YT,FCALL,RVEC,IGEN,
     +            ISEED,RWSAV,IFLAG)
      FCALL = .FALSE.
      CALL G05HKF(DIST,NUM,IP,IQ,PARAM,GAMMA,DF,HT,YT,FCALL,RVEC,IGEN,
     +            ISEED,RWSAV,IFLAG)
      CALL G05HKF(DIST,NUM,IP,IQ,PARAM,GAMMA,DF,HT,YT,FCALL,RVEC,IGEN,
     +            ISEED,RWSAV,IFLAG)
```

```
      IFLAG = -1
      DO 160 I = 1, NUM
         XTERM = ZERO
         DO 140 K = 1, NREG
            XTERM = XTERM + X(I,K)*BX(K)
  140    CONTINUE
         IF (MN.EQ.1) THEN
            YT(I) = MEAN + XTERM + YT(I)
         ELSE
            YT(I) = XTERM + YT(I)
         END IF
  160 CONTINUE

      COPTS(1) = .TRUE.
      COPTS(2) = .TRUE.

      MAXIT = 200
      TOL = 1.0e-5

      DO 180 I = 1, NPAR
         THETA(I) = PARAM(I)*0.5e0
  180 CONTINUE

      THETA(NPAR+ISYM) = GAMMA*0.5e0
      THETA(NPAR+ISYM+1) = DF*0.65e0

      NPAR2 = 2 + IP + IQ + ISYM + MN + NREG
      CALL G13FAF(DIST,YT,X,LDX,NUM,IP,IQ,NREG,MN,ISYM,NPAR2,THETA,SE,
     +            SC,COVAR,NPARMX,HP,ETM,HTM,LGF,COPTS,MAXIT,TOL,WK,LWK,
     +            IFLAG)

      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Student t-distribution'
      WRITE (NOUT,*)
      WRITE (NOUT,*) '          Parameter       Standard        Correct'
      WRITE (NOUT,*) '          estimates       errors          values'

      DO 200 I = 1, NPAR
         WRITE (NOUT,99999) THETA(I), SE(I), PARAM(I)
  200 CONTINUE

      IF (ISYM.EQ.1) THEN
         WRITE (NOUT,99999) THETA(NPAR+ISYM), SE(NPAR+ISYM), GAMMA
      END IF

      WRITE (NOUT,99999) THETA(NPAR+ISYM+1), SE(NPAR+ISYM+1), DF

      IF (MN.EQ.1) THEN
         WRITE (NOUT,99999) THETA(NPAR+ISYM+1+MN), SE(NPAR+ISYM+1+MN),
     +     MEAN
      END IF

      DO 220 I = 1, NREG
         WRITE (NOUT,99999) THETA(NPAR+ISYM+1+MN+I),
     +     SE(NPAR+ISYM+1+MN+I), BX(I)
  220 CONTINUE

      NT = 4
      CALL G13FBF(NUM,NT,IP,IQ,THETA,GAMMA,CVAR,HTM,ETM,IFLAG)

      WRITE (NOUT,*)
      WRITE (NOUT,99998) 'Volatility forecast = ', CVAR(NT)
      STOP
*
99999 FORMAT (1X,3F16.4)
99998 FORMAT (1X,A,F12.4)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
G13FAF Example Program Results

Normal distribution

          Parameter       Standard        Correct
          estimates       errors          values
          0.8031          0.0788          0.8000
          0.6249          0.0570          0.6000
          0.1803          0.0327          0.2000
          0.0921          0.0237          0.1000
         -0.5119          0.0682         -0.4000
          2.9860          0.0324          3.0000

Volatility forecast =       2.8040


Student t-distribution

          Parameter       Standard        Correct
          estimates       errors          values
          0.0871          0.0244          0.1000
          0.2174          0.0549          0.2000
          0.2736          0.0849          0.3000
          0.3588          0.0811          0.4000
         -0.3240          0.0650         -0.2000
          4.5173          0.5238          4.1000
          3.0182          0.0431          3.0000
          1.4727          0.0269          1.5000
          2.4640          0.0319          2.5000

Volatility forecast =       0.4133
```